

Description

`collect style header` edits the content of the table headers. With this style, you can specify how the dimensions and levels are displayed in row, column, and table headers. For each dimension, the name of the dimension, the label of the dimension, or nothing may be displayed. Likewise, for levels within a dimension, the label of that level, the value of the level, or nothing may be displayed.

Quick start

For all dimensions, display the dimension label and the level labels in row, column, and table headers

```
collect style header, title(label) level(label)
```

For dimension `d1`, hide the dimension title

```
collect style header d1, title(hide)
```

For dimension `d1`, display the level values

```
collect style header d1, level(value)
```

Menu

Statistics > Summaries, tables, and tests > Tables and collections > Build and style table

Syntax

```
collect style header [ dimlist ] [ , options ]
```

dimlist is a dimension in a collection, a dimension and its corresponding level or list of levels, or a combination of these:

- dimension*
- dimension* [*level*]
- dimension* [*level1 level2 ...*]

| <i>options</i> | Description |
|--|---|
| <code>name(<i>cname</i>)</code> | specify header styles for collection <i>cname</i> |
| <code>title(<i>tstyle</i>)</code> | specify dimension title header style |
| <code>level(<i>lstyle</i>)</code> | specify dimension level header style |
| <code>nformat(<i>%fmt</i>)</code> | specify numeric format for numeric levels |
| <code>fvlevels(<i>fvlstyle</i>)</code> | specify factor-variable level header style |
| <code>basestyle</code> | update basestyle properties |

`basestyle` does not appear in the dialog box.

Options

`name(cname)` specifies the collection to which header style properties are to be applied. By default, properties are applied to the current collection.

`title(tstyle)` specifies the dimension title header style to be used in row, column, and table headers. *tstyle* may be `label`, `name`, or `hide`.

`label` specifies that `collect` use the dimension's label for headers. If a dimension does not have a label, then `collect` will use the dimension's name.

`name` specifies that `collect` use the dimension's name for headers.

`hide` specifies that `collect` not show the dimension's label or name in the headers.

The default is `title(hide)`.

`level(lstyle)` specifies the dimension's level header style to be used in row, column, and table headers. *lstyle* may be `label`, `value`, or `hide`.

`label` specifies that `collect` use the level's value labels for headers. If a level does not have a label, then `collect` will use the level's value.

`value` specifies that `collect` use the level's values for headers.

`hide` specifies that `collect` not show the level's labels or values in the headers.

The default is `level(label)`.

`nformat(%fmt)` applies the Stata numeric format *%fmt* to numeric levels for a given dimension in the header.

`fvlevels(fvstyle)` specifies the dimension's factor-variable level header style to be used in row, column, and table headers. *fvstyle* may be `show` or `hide`.

`show` specifies that `collect` show factor-variable levels in the headers.

`hide` specifies that `collect` not show factor-variable levels in the headers.

The default is `fvlevel(show)`.

`basestyle` indicates that the header style edits be applied to the base header style properties, instead of overriding the current style for the headers.

Each header begins with baseline header style properties. (You can view your table with these baseline properties by first clearing out the collection styles with `collect style clear`.) The appearance of the headers is then updated with any changes specified in the default style used by `collect` and `table`. Any `collect style header` command you issue will override the current style for that header. If you specify the `basestyle` option, the style changes will instead apply to the baseline style and they will not override any current header style edits targeted to specific dimensions.

For example, suppose your current style displays values for levels of dimensions. You then decide to display the labels for the levels of the dimension `result`, by typing `collect style header result, level(label)`. After you preview your table, you choose to hide the values and labels for all other dimension levels. You can type `collect style header, level(hide) basestyle` to make this change while continuing to display the labels for levels of the dimension `result`.

Remarks and examples

collect style header specifies the way that dimensions and their levels be displayed in row, column, and table headers. collect style header is often used in combination with `collect label dim` and `collect label levels` to get the desired wording in the headers.

To demonstrate, we first collect results using the `collect` prefix and lay out a table using `collect layout`.

```
. use https://www.stata-press.com/data/r19/nhanes2
. quietly: collect _r_b: regress bpsystol bmi
. quietly: collect _r_b: regress bpsystol bmi age
. collect layout (colname) (cmdset#result)
Collection: default
  Rows: colname
  Columns: cmdset#result
Table 1: 3 x 2
```

| | 1 | 2 |
|-----------------------|-------------|-------------|
| | Coefficient | Coefficient |
| Body mass index (BMI) | 1.656894 | 1.304128 |
| Age (years) | | .5883367 |
| Intercept | 88.56855 | 69.58451 |

By default, we do not see names or labels for the dimensions. However, we do see the labels for all levels that are labeled—the variable labels are the labels for the levels of `colname`, and `Coefficient` is the label for the `_r_b` level of the dimension `result`. The levels of the `cmdset` dimension do not have labels, so we see the values of these levels.

Because the coefficient is the only statistic in the table, we could hide its label by specifying the `level(hide)` option for the `result` dimension.

```
. collect style header result, level(hide)
. collect preview
```

| | 1 | 2 |
|-----------------------|----------|----------|
| Body mass index (BMI) | 1.656894 | 1.304128 |
| Age (years) | | .5883367 |
| Intercept | 88.56855 | 69.58451 |

If the levels of `cmdset` had labels, they would show because `level(label)` is the default for all dimensions. Here we add labels to the levels of this dimension, and they automatically appear in the column headers.

```
. collect label levels cmdset 1 "Model 1" 2 "Model 2"
. collect preview
```

| | Model 1 | Model 2 |
|-----------------------|----------|----------|
| Body mass index (BMI) | 1.656894 | 1.304128 |
| Age (years) | | .5883367 |
| Intercept | 88.56855 | 69.58451 |

Suppose we wanted to see the names of our variables (the values of the levels of the `colname` dimension) rather than their labels on the rows. We can request this with option `level(value)`. We can also specify a new label for the `colname` dimension and show this label in the row headers by specifying the `title(label)` option with `collect style header`.

```
. collect label dim colname "Covariates", modify
. collect style header colname, level(value) title(label)
. collect preview
```

| | Model 1 | Model 2 |
|------------|----------|----------|
| Covariates | | |
| bmi | 1.656894 | 1.304128 |
| age | | .5883367 |
| _cons | 88.56855 | 69.58451 |

Let’s add estimation results using a factor variable.

```
. quietly: collect _r_b: regress bpsystol bmi age i.female
. collect label values cmdset 3 "Model 3"
. collect preview
```

| | Model 1 | Model 2 | Model 3 |
|------------|----------|----------|-----------|
| Covariates | | | |
| bmi | 1.656894 | 1.304128 | 1.305773 |
| age | | .5883367 | .589271 |
| Male | | | 0 |
| Female | | | -4.063319 |
| _cons | 88.56855 | 69.58451 | 71.63197 |

Recall that dimension titles are hidden by default. This applies to factor variables, too, so we see the value labels of the `female` variable but not its name or label. Because `female` is binary, we can hide the base level and show its name instead of the value labels. We use `collect style showbase off` to hide the base level. We then use `collect style header` to show `female`’s variable name in the header’s title and to hide the factor-variable levels in the header for dimension `colname`.

```
. collect style row stack, nobinder
. collect style showbase off
. collect style header female, title(name)
. collect style header colname, fvlevels(hide)
. collect preview
```

| | Model 1 | Model 2 | Model 3 |
|------------|----------|----------|-----------|
| Covariates | | | |
| bmi | 1.656894 | 1.304128 | 1.305773 |
| age | | .5883367 | .589271 |
| female | | | -4.063319 |
| _cons | 88.56855 | 69.58451 | 71.63197 |

The `fvlevels()` style property works on interactions, too, and is used in the `anova` predefined style; see [TABLES] [Predefined styles](#) and [TABLES] [Example 8](#).

Here we add estimation results using an interaction and preview the table using the same header styles as in the previous table.

```
. quietly: collect _r_b: regress bpsystol bmi c.age##i.female
. collect label values cmdset 4 "Model 4"
. collect preview
```

| | Model 1 | Model 2 | Model 3 | Model 4 |
|--------------|----------|----------|-----------|-----------|
| Covariates | | | | |
| bmi | 1.656894 | 1.304128 | 1.305773 | 1.264658 |
| age | | .5883367 | .589271 | .4390171 |
| female | | | -4.063319 | -17.78472 |
| female # age | | | | .288516 |
| _cons | 88.56855 | 69.58451 | 71.63197 | 79.8064 |

In the examples above, we have modified our header styles for a selected dimension. However, collect style header is not limited to modifying only one dimension. If we wish to make a change for all dimensions, we can simply omit the dimension names from the command. For instance, we could type

```
. collect style header, title(label)
```

Alternatively, we could specify a header style for multiple dimensions. For instance, we could type

```
. collect style header cmdset colname, title(label)
```

If you have a preferred method of displaying the dimensions and their levels for many of the tables you create, you can use `collect style save` to save a file with this style along with any others you like. Then, with future collections, you can use `collect style use` to apply this header style to future collections and tables.

Stored results

collect style header stores the following in `s()`:

```
Macros
      s(collection)  name of collection
```

Also see

- [TABLES] `collect label` — Manage custom labels in a collection
- [TABLES] `collect query` — Query collection style properties
- [TABLES] `collect style save` — Save collection styles to disk
- [TABLES] `collect style use` — Use collection styles from disk

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

